***********************************************************************************************
*Table A13: Robots and college enrollment: Alternative construction of robot exposure measures*
***********************************************************************************************

*Education
use "$final_data_outcomes/czone_Students.dta", clear
merge 1:1 czone year using "$final_data_outcomes/IPEDS_EnrollmentCZ.dta", assert(1 3) nogenerate

*Automation
merge 1:1 czone year using "$final_data_automation/czones_ExposureAutomation.dta", assert(1 3) nogenerate

*Imports merge
merge 1:1 czone year using "$final_data_imports/czone_ExposureChina", assert(1 3) nogenerate

*Covariates
merge m:1 czone using "$final_data_covariates/czone1990_Covariates.dta", assert(3) nogenerate //beginning of period
merge m:1 czone using "$final_data_covariates/IPEDS_institutionControlsCZ.dta", assert(1 3) nogenerate

*Keep only periods of interest
keep if year<=2000

xtset czone year

foreach var in public private uniTop30Dummy largest150 anyAid {
replace `var' = 0 if mi(`var')
}

egen medianAnyAid = median(anyAid), by(statefip)
	replace anyAid = medianAnyAid if mi(anyAid)

eststo clear
*Euro 8 countries (include also Germany)
eststo: ivreghdfe d_sc (expof_us_adj d_exp_us_stacked = expof_euro8_qo d_exp_iv_stacked) d_sc_70_90 $occupations_1990 $demographics_1990 $industry_shares_1990 $institutions_1990 [w=ipums_pop_1990], absorb(division##year statefip) cluster(statefip)

*Euro 5 countries (Acemoglu and Restrepo measure)
eststo: ivreghdfe d_sc (expof_us_adj d_exp_us_stacked = expof_euro5_qo d_exp_iv_stacked) d_sc_70_90 $occupations_1990 $demographics_1990 $industry_shares_1990 $institutions_1990 [w=ipums_pop_1990], absorb(division##year statefip) cluster(statefip)

*No adjustment term
eststo: ivreghdfe d_sc (expof_us_lv d_exp_us_stacked = expof_euro7_lv d_exp_iv_stacked) d_sc_70_90 $occupations_1990 $demographics_1990 $industry_shares_1990 $institutions_1990 [w=ipums_pop_1990], absorb(division##year statefip) cluster(statefip)

*Shares of 1990 also for instrument
eststo: ivreghdfe d_sc (expof_us_adj d_exp_us_stacked = expof_b90_euro7_qo d_exp_iv_stacked) d_sc_70_90 $occupations_1990 $demographics_1990 $industry_shares_1990 $institutions_1990 [w=ipums_pop_1990], absorb(division##year statefip) cluster(statefip)

estout using "$tables_appendix/Table A12.tex", style(tex) rename(expof_us_lv expof_us_adj) ///
varlabels(expof_us_adj "US robot exposure") ///
cells(b(fmt(%9.3f) star) se(par)) nolabel replace mlabels(none) collabels(none) starlevels(\$^{*}\$ .1 \$^{**}\$ .05 \$^{***}\$ .01) /// 
keep(expof_us_adj) ///
order(expof_us_adj)
